clear
set mem 500m
set more off
set logtype text
set matsize 3000

cap log close
log using "$Log/Table6.log", replace


use "$AnalysisData/bills_analysis_101-111_replication.dta"
keep if private==0

bysort v2 sponsor_state_abbrev sponsor_district sponsor_term_served: egen tot_bills=count(v2)

gen byte sponsor_democrat = sponsor_party==100 if sponsor_party~=.

gen byte NE = sponsor_state_icpsr>=1 & sponsor_state_icpsr<=14 if sponsor_state_icpsr~=.
gen byte MW = sponsor_state_icpsr>=21 & sponsor_state_icpsr<=37 if sponsor_state_icpsr~=.
gen byte SO = sponsor_state_icpsr>=40 & sponsor_state_icpsr<=56 if sponsor_state_icpsr~=.
gen byte WE = sponsor_state_icpsr>=61 & sponsor_state_icpsr<=82 if sponsor_state_icpsr~=.

replace sponsor_age=. if sponsor_age>100
gen sponsor_rookie=(sponsor_tenure_run==1) if sponsor_tenure_run!=.
gen byte leader = (comc==1) | (comr==1) if comc~=. & comr~=. 

gen lnpden=lnpop-lnarea
gen absMV=abs(MV1_democrat)

egen group_congressmen = group(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)
egen tag_sponsor = tag(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)
egen group_sponsor = group(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)

rename numb_cosponsors_tenure_59 numb_cosponsors_ten_59
rename numb_cosponsors_tenure_10plus numb_cosponsors_ten_10p
egen tag_congressmen=tag(v2 sponsor_state_abbrev sponsor_district sponsor_term_served)

cap drop pct_cosponsors_fem
gen pct_cosponsors_fem=100*numb_cosponsors_fem/(numb_cosponsors+1)
replace pct_cosponsors_fem=. if pct_cosponsors_fem>100

cap drop  pct_cosponsors_leader
gen pct_cosponsors_leader=100*numb_cosponsors_leader/(numb_cosponsors+1)
replace pct_cosponsors_leader=. if pct_cosponsors_leader>100

gen numb_cosponsors_ten_5p=numb_cosponsors_ten_59+numb_cosponsors_ten_10p
gen pct_cosponsors_ten_5p=100*numb_cosponsors_ten_5p/(numb_cosponsors+1)
replace pct_cosponsors_ten_5p=. if pct_cosponsors_ten_5p>100

compress

egen tag_bill = tag(v2 HRnumber)

gen int decade = 1980 if v2<=102
replace decade=1990 if v2>=103 & v2<=107
replace decade=2000 if v2>=108

egen district_id = group(decade sponsor_state_abbrev sponsor_district)
 
gen maj_house=0
replace maj_house=1 if v2==101 & sponsor_party==100
replace maj_house=1 if v2==102 & sponsor_party==100
replace maj_house=1 if v2==103 & sponsor_party==100
replace maj_house=1 if v2==104 & sponsor_party==200
replace maj_house=1 if v2==105 & sponsor_party==200
replace maj_house=1 if v2==106 & sponsor_party==200
replace maj_house=1 if v2==107 & sponsor_party==200
replace maj_house=1 if v2==108 & sponsor_party==200
replace maj_house=1 if v2==109 & sponsor_party==200
replace maj_house=1 if v2==110 & sponsor_party==100
replace maj_house=1 if v2==111 & sponsor_party==100
replace maj_house=1 if v2==112 & sponsor_party==200
replace maj_house=1 if v2==113 & sponsor_party==200
 
local controls1 = "i.v2 "
local controls2 = "i.v2 MV1_female femaleXMV1_female "
local controls3 = "i.v2 MV1_female femaleXMV1_female "
local controls4 = "i.v2 "
local controls5 = "i.v2 "

local other_controls = "i.minor house_* sponsor_democrat sponsor_rookie sponsor_tenure_run sponsor_age leader ivycoll black occ0 occ1 occ2 occ3 occ4 borninstate tot_bills NE MW WE pct_black pct_urban pct_for_born pct_age_over65 lninc lnpden"
local billchars = "i.minor house_*"
local indivchars = "sponsor_democrat sponsor_rookie sponsor_tenure_run sponsor_age leader ivycoll black occ0 occ1 occ2 occ3 occ4 borninstate tot_bills"
local districtchars = "NE MW WE pct_black pct_urban pct_for_born pct_age_over65 lninc lnpden"

local if1 = "if tag_bill==1"
local if2 = "if sponsor_party==100 & tag_bill==1"
local if3 = "if sponsor_party==200 & tag_bill==1"

foreach var of varlist var_cosp_dwnom1_spons m_betweenness_w_spons {
   replace `var'=`var'*1000
}



foreach depvar of varlist pct_cosponsors_fem pct_cosponsors_leader pct_cosponsors_ten_5p m_betweenness_w_spons var_cosp_dwnom1_spons {
   forvalues j=1/5 {
      forvalues k = 1/3 {
	     qui reg `depvar' sponsor_female `controls3' `other_controls' `if`k'' & mixed_gender_election==1, robust cluster(group_sponsor)
	     qui gen sample=e(sample)

         di ""
	     di ""
	     di "j = ", `j', "k = ", `k'
	     di ""
	     di ""
	     
   	     di "rdbwselect_2014 `depvar' MV1_female `if`k'', c(0) kernel(uniform)"
	     rdbwselect_2014 `depvar' MV1_female `if`k'', c(0) kernel(uniform)
	     local band = e(h_CCT)

	     if `j'==1 {
	        di "reg `depvar' sponsor_female `controls`j'' `other_controls' `if`k'', robust cluster(group_sponsor)"
	        reg `depvar' sponsor_female `controls`j'' `other_controls' `if`k'', robust cluster(group_sponsor)   
	     }
	     else if `j'==2 {
	        di "reg `depvar' sponsor_female `controls`j'' `if`k'' & sample==1 & abs(MV1_female)<=`band', robust cluster(group_sponsor)"
	        reg `depvar' sponsor_female `controls`j'' `if`k'' & sample==1 & abs(MV1_female)<=`band', robust cluster(group_sponsor)   
	     }
	     else if `j'==3 {
            qui probit sponsor_female i.v2 `districtchars' `if`k'' & tag_sponsor==1 & abs(MV1_female)<=`band'
            predict pscore if sample==1 
            gen wt = 1/pscore if sponsor_female==1
            replace wt =1/(1-pscore) if sponsor_female==0
	        di "reg `depvar' sponsor_female `controls`j'' `billchars' [aw=wt] `if`k'' & sample==1 & abs(MV1_female)<=`band', robust cluster(group_sponsor)"
            reg `depvar' sponsor_female `controls`j'' `billchars' [aw=wt] `if`k'' & sample==1 & abs(MV1_female)<=`band', robust cluster(group_sponsor)
            drop pscore wt
         }
	     else if `j'==4 {
            qui probit sponsor_female i.v2 `districtchars' absMV `if`k'' & tag_sponsor==1 
            predict pscore
            gen wt = 1/pscore if sponsor_female==1
            replace wt =1/(1-pscore) if sponsor_female==0
	        di "reg `depvar' sponsor_female `controls`j'' `billchars' [aw=wt] `if`k'', robust cluster(group_sponsor)"
            reg `depvar' sponsor_female `controls`j'' `billchars' [aw=wt] `if`k'', robust cluster(group_sponsor)
            drop pscore wt
         }
	     else if `j'==5 {
            qui probit sponsor_female i.v2 `districtchars' `indivchars'  absMV `if`k'' & tag_sponsor==1 
            predict pscore
            gen wt = 1/pscore if sponsor_female==1
            replace wt =1/(1-pscore) if sponsor_female==0
	        di "reg `depvar' sponsor_female `controls`j'' `billchars' [aw=wt] `if`k'', robust cluster(group_sponsor)"
            reg `depvar' sponsor_female `controls`j'' `billchars' [aw=wt] `if`k'', robust cluster(group_sponsor)
            drop pscore wt
         }

		  if `j'~=2 & `j'~=3 {
	         scalar b_col`j'_row`k' = _b[sponsor_female]
	         scalar se_col`j'_row`k' = _se[sponsor_female]
	         scalar n_col`j'_row`k' = e(N)
			 sum tag_congressmen if tag_congressmen==1 & e(sample)
			 scalar ni_col`j'_row`k' = e(N_clust)
	         scalar ob_col`j'_row`k' = .
		  }
		  else if `j'==2 | `j'==3 {
	         scalar b_col`j'_row`k' = _b[sponsor_female]
	         scalar se_col`j'_row`k' = _se[sponsor_female]
	         scalar n_col`j'_row`k' = e(N)
			 sum tag_congressmen if tag_congressmen==1 & e(sample)
			 scalar ni_col`j'_row`k' = e(N_clust)
	         scalar ob_col`j'_row`k' = round(`band')
		  }
         drop sample
      
      }
   }

preserve

* Display results

forvalues i = 1/5 {
   gen var`i' =.
}
gen str20 var6 = ""
		  
forvalues j=1/5 {
   forvalues k = 1/3 {
      replace var`j' = b_col`j'_row`k' if _n==6*(`k'-1)+1
      replace var`j' = se_col`j'_row`k' if _n==6*(`k'-1)+2
      replace var`j' = n_col`j'_row`k' if _n==6*(`k'-1)+3
      replace var`j' = ni_col`j'_row`k' if _n==6*(`k'-1)+4
      replace var`j' = ob_col`j'_row`k' if _n==6*(`k'-1)+5
   }
}   

keep var1-var6
keep if _n<=18

order var6 var1-var5

ren var6 sampletype
ren var1 OLS_all
ren var2 RD_bwidth
ren var3 RD_match_bw
ren var4 PS_match
ren var5 PS_match_indiv

foreach var of varlist OLS_all RD_bwidth RD_match_bw PS_match PS_match_indiv {
   gen str3 `var'_stars ="*" if abs(`var'/`var'[_n+1])>=1.645 & mod(_n,6)==1
   replace `var'_stars ="**" if abs(`var'/`var'[_n+1])>=1.96 & mod(_n,6)==1
   replace `var'_stars="***" if abs(`var'/`var'[_n+1])>=2.58 & mod(_n,6)==1
}

replace sampletype = "All" if _n>=1 & _n<=5
replace sampletype = "Democrats" if _n>=7 & _n<=11
replace sampletype = "Republicans" if _n>=13 & _n<=17

order sampletype OLS_all OLS_all_stars RD_bwidth RD_bwidth_stars RD_match_bw RD_match_bw_stars PS_match PS_match_stars PS_match_indiv PS_match_indiv_stars 
    export excel using "$Output/Table6_`depvar'", firstrow(var) replace
restore
}		  
	  	
log close